


# State\_Variable\_OTAs\_100Hz

\* dsauersanjose@aol.com  
 \* www.idea2ic.com  
 \* 4/15/08



$$H(s) = 1/(s^2 + s*A1 + A0)$$

$$H(s) = 1/(s^2 + s*w0/Q + w0^2)$$

Set **A0 = 1** and scale **s** to **1KHz**  
 Then **R2 = Q** and **s = 2\*PI\*1KHz**


```

.OPTIONS GMIN=1e-12 METHOD=trap srcsteps = 1 gminsteps = 1
=====
V_IN     VIN     0     AC     1     DC     0
R0       VIN     SUMN   10k
R1       SUMN    HP     10k
R2       SUMN    BP     100k
R3       SUMN    LP     10k
XOPA1    SUMN    0      HP     OPA
XOTAS1   HP      BP     OTA_S
XOTAS2   BP      LP     OTA_S
.ac      dec     50     10     10k
  
```


==OTAs\_Make\_It\_Possible\_To\_Voltage\_Control\_Frequency==

```

.control
run
plot db(bp) db(hp) db(lp) title StateVariable_Q_10
alter R2 resistance = 10k
run
plot db(bp) db(hp) db(lp) title StateVariable_Q_1
.endc
  
```



[http://www.idea2ic.com/PlayWithJavascript/R\\_C\\_Freq.html](http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html)



```

.SUBCKT OPA INP INN OUT
EGN1 O1 0 INP INN -1
EGN2 OUT 0 IP 0 -1000000
R1 O1 IP 10k
C1 OUT IP 10p

```

.ends

```

.SUBCKT OTA_S IN OUT
QN1 VC1 IN VE1 NPNP
QN2 VC2 0 VE1 NPNP
QP1 VC1 VC1 VCC PNPP
QP2 VC2 VC1 VCC PNPP
IB VE1 0 .52u
VCC VCC 0 DC 2
EGN2 OUT 0 VC2 0 +1
C1 VC2 0 .01592u

```

.ends

```


.model NPNP NPN( BF=2100 VAF=216 )
.model PNPP PNP( BF=2100 VAF=210 )

```

.end

=====END\_OF\_SPICE=====

The effective resistance of OTAs is controllable.  
Voltage Controlled filters can be made using OTAs.



Since Q is independently controlled by R1-R3  
The Voltage Controlled filter can have  
a 6 pole bessel filter response over a wide  
frequency range.

